Maths with Maple — Maple reference

1. Generalities

(1.1) For help, type `?` and press ENTER. (Alternatively, you can click where it says "Help" at the top right of the Maple window.) For help on plotting (for example) you can type `?plot` or `help(plot);` and press ENTER. Alternatively, you can click on "Help", then select "Topic search" or "Full text search" from the resulting menu, then enter "plotting" in the search box.

(1.2) See Section 14 for advice on some common problems.

(1.3) Commands must end with a semicolon (or occasionally a colon). If you see the message "Warning, premature end of input", then you have probably left out the semicolon.

(1.4) If you use a colon then Maple will carry out your command but will not tell you the answer; this is sometimes useful when calculating intermediate results that would take a lot of space to print.

(1.5) You can use the symbol `%` to refer to the last thing that Maple calculated, and `%%` to refer to the one before that, and so on. Even if you use a colon to prevent Maple from displaying a result, you can still use `%` to refer to that result.

(1.6) If you move around a worksheet inserting things in different places, you should remember that `%` refers to the **most recent** result, which may not be the one that you see directly above the line where you are typing.

(1.7) It is important to understand the difference between `=` and `:=`. If we enter $x := 3;$, this tells Maple that x is definitely equal to 3, so Maple will replace x by 3 whenever it sees an x. If instead we enter $x = 3$, we are mentioning the question of whether x might be 3, or temporarily considering the case when $x = 3$. This has no effect in itself, until it is used in a command like `solve(x=3)` or `eval(x^2+5,x=3)` or `limit((x^2-9)/(x-3),x=3)`.

(1.8) If you want to define a function (say $f(x) = x^2 + 3$) it does *not* work to enter `f(x):=x^2+3;`. See Section 9.

(1.9) Enter `restart;` to restart Maple, removing any values that you may have assigned to variables. You should generally do this at the beginning of each new exercise, otherwise stray values left over from previous exercises can cause trouble and confusion. You can also restart by clicking the button with a circulating arrow at the right-hand end of the Maple toolbar.

(1.10) If you just want to clear a single variable (say x) you can enter `unassign('x')` instead. Note the single quote marks around the x.

2. Numerical approximation

(2.1) To get an explicit numerical approximation for an expression, use the `evalf()` function. For example, `evalf(Pi)` gives 3.141592654, and `evalf(x/3)` gives $0.3333333333x$.

(2.2) The `evalf()` function generally gives an answer to 10 significant figures. If you want to calculate $\pi^2/6$ to 50 digits (for example), you can enter `evalf[50](Pi^2/6)`. Similarly, `evalf[12](1/7)` gives 1/7 to 12 decimal places.

(2.3) As an alternative, you can enter `Digits:=100;` to tell Maple to use 100 digits in all calculations until further notice. This approach is usually a better choice, as it ensures that the extra digits will be kept in subsequent calculations. You may assume that `Digits:=100` will give you exactly 100 significant figures, but the real truth is a little more complicated than that.

3. Notation for variables

(3.1) Most variables will have single-letter names, such as x or A.

(3.2) Note that case is significant: Maple regards A as different from a, and x as different from X. Similarly, $\sin(x)$ must be entered as `sin(x)` and not `Sin(x)`; this is a fairly common source of trouble.

(3.3) It is perfectly permissible to use long names for variables, such as `eqns` or `sols` or `SpeedOfLight`.

(3.4) However, you may not use words that already have a special meaning in Maple. In particular, you cannot call a variable `roots` or `name`, for example.

(3.5) Greek letters can be entered using their english names. For example, if you enter `theta+phi` then Maple will print it as $\theta + \phi$. However, the constant $\pi \simeq 3.14159$ must be entered as `Pi` (with a capital P) not `pi` (see 6.2).

(3.6) Sometimes one uses subscripted variables, such as x_3 or $a_{2,4}$ or Solution_5. These should be entered as `x[3]` or `a[2,4]` or `Solution[5]`.

4. Algebraic operations

(4.1) Use a `*` for multiplication; for example, $2xy$ should be entered as `2*x*y`, and $(2x+1)\sin(3x)$ should be entered as `(2*x+1)*sin(3*x)`.

(4.2) An error message saying "Error, missing operator or ';'" often indicates a missing `*`. Things like `2(x+y)` (with a number followed by an open bracket) are particularly dangerous; Maple will (for technical reasons that make some sense in the right context) silently convert `2(x+y)` to 2 without any error message.
(4.3) Note also that Maple will treat `xy` as a two-letter name for a single variable; if you want x times y, you must enter `x*y`.
(4.4) Just as in standard mathematical notation, bracketing is important. To multiply $x+2$ by $x-2$, you must enter `(x+2)*(x-2)`, not `x+2*x-2`.
(4.5) Additionally, if you want to multiply by something that starts with a minus sign, you must enclose it in brackets; for example, enter `(x+y)*(-u)`, not `(x+y)*-u`.
(4.6) Enter powers using the `^` character, for example `x^6` for x^6.
(4.7) Just as in standard mathematical notation, bracketing is important. To raise $x+2$ to the power $n+5$, you must enter `(x+2)^(n+5)`, not `x+2^n+5`. To get the cube root of x, enter `x^(1/3)`, not `x^1/3`.
(4.8) Additionally, if you want to raise something to a power that starts with a minus sign, you must enclose it in brackets; for example, enter `(x+y)^(-2)`, not `(x+y)^-2`.
(4.9) The square root of x can be entered as `sqrt(x)` or `x^(1/2)`. Similarly, $\sqrt{b^2-4ac}$ is `sqrt(b^2-4*a*c)`. For more general roots you should use power notation, such as `(x+y)^(1/4)` for the fourth root of $x+y$.

5. Algebraic manipulation

Suppose we have a complicated expression called A, which we want to simplify or otherwise manipulate.

(5.1) To substitute for some or all of the variables in A, use the `subs` command. For example, if A is $(a+b)^n$, you can change the b to 1 and the n to p/q by entering `subs(b=1,n=p/q,A)`, which gives $(a+1)^{p/q}$. The syntax `subs({b=1,n=p/q},A)` will also work.
(5.2) Often the equations to be used in `subs(...)` will come from the `solve(...)` command. For example, to find the value of x^2+y^2 at the point where $x+y=6$ and $x-y=2$, enter `sol:=solve({x+y=6,x-y=2},{x,y})` and then `subs(sol,x^2+y^2)`.
(5.3) Enter `expand(A)` to expand everything out. For example, enter `expand((x+y)^3)` to expand $(x+y)^3$, giving $x^3+3x^2y+3xy^2+y^3$.
(5.4) To simplify A, try entering `simplify(A)`. If this does not do what you want, you can try `factor(A)`. If A involves trigonometric functions, try `simplify(expand(convert(A,exp)))`.
(5.5) If Maple is refusing to simplify $(a^4)^{1/4}$ to a (for example), you can try `simplify(A,symbolic)`. You should be aware that the answer **may not be correct** in all cases, because $(a^4)^{1/4}$ really is different from a when $a<0$. Similarly, the simplification $\ln(e^x)=x$ is not always valid when x is a complex number, so `simplify(ln(exp(x)))` just gives $\ln(e^x)$, but `simplify(ln(exp(x)),symbolic)` gives x.
(5.6) If you suspect that A is equal to some other expression (say B) and want to check this, the best way is to enter `simplify(A-B)` and see if you get zero.
(5.7) If Maple writes an expression with the terms in an unnatural order (eg $t^3+t^4+t^2+t$ instead of $t^4+t^3+t^2+t$) then use the `sort` command to rearrange them.
(5.8) To collect together all the terms in A involving the same power of x, enter `collect(A,x)`. For example, this converts $1+ax+bx+cx^2+dx^2$ to $1+(a+b)x+(c+d)x^2$.
(5.9) If you just want the coefficient of x^2 then you can instead enter `coeff(A,x,2)` or `coeff(A,x^2)`. To get the constant term, you must enter `coeff(A,x,0)` — the syntax `coeff(A,x^0)` will not work.

6. Constants

(6.1) The constant $\pi \simeq 3.1415926536$ should be entered as `Pi`, with a capital P. Maple will generally leave this as a symbol; if you want the numerical value, you should use the `evalf()` function.
(6.2) If you enter `pi` with a small p, then Maple will accept it without a syntax error and will print it as π, but will not treat it as mathematically equivalent to `Pi`. If you have an expression like $\sin(\pi)$ and Maple stubbornly refuses to convert it to zero, then this may be the cause; you may have entered `pi` instead of `Pi`. You can cure this by entering `pi:=Pi;`, which will tell Maple to treat `pi` the same as `Pi`.
(6.3) The constant $e \simeq 2.718281828$ **cannot** (by default) be entered as `e`. Instead, enter `exp(1)` for e, and `exp(x)` for e^x. If you want Maple to treat `e` as meaning `exp(1)`, you can enter `e:=exp(1);`.
(6.4) The square root of minus one can be entered as `I` (not `i` or `j`).

7. Solving

(7.1) To solve the equation $x^2-5x+6=0$ (for example), enter `solve(x^2-5*x+6=0,{x})`. Similarly, if you have already defined a function g, you can enter `solve(g(t)=0,{t})` to solve the equation $g(t)=0$.

(7.2) The above syntax, with curly brackets around the `x`, gives the answer in the form $\{x = 2\}, \{x = 3\}$. If you prefer to have the answer in the form $2, 3$ (with no "$x =$") , you should instead enter `solve({x^2-5*x+6=0},x)` (without curly brackets around the `x`).

(7.3) To solve several equations in several variables, put the equations in one set of curly brackets, and the variables in another set. For example, to solve $2x + 3y = 5$ and $3x + 4y = 7$ for x and y, enter

```
solve({2*x+3*y=5,3*x+4*y=7},{x,y});
```

(7.4) If there are no solutions, Maple will just give you a new prompt without printing any answer at all.

(7.5) If the answer involves the word `RootOf`, it may help to enter `_EnvExplicit:=true;` and try again.

(7.6) If the problem involves trigonometric functions, then there will typically be infinitely many solutions, but Maple will only list a few of them. To get all solutions, enter `_EnvAllSolutions:=true;`.

(7.7) To get approximate numerical solutions, use `fsolve` instead of `solve`. This will generally only find one solution (unless the equation to be solved is very simple).

(7.8) To find more solutions, you must tell `fsolve` where to start looking. For example, `fsolve({sin(x)=0},{x=3})` finds a root of $\sin(x)$ near $x = 3$.

(7.9) As with the `solve()` command, if you are solving for a single variable, you can get the answer in two different forms. For example, `fsolve({x^2=2},x)` gives $-1.414213562, 1.414213562$, whereas `fsolve({x^2=2},{x})` gives $\{x = -1.414213562\}, \{x = 1.414213562\}$.

(7.10) Sometimes we need to solve problems like this: find a and b such that $a(x-1)^2 + b(x+1)^2 = x$ *for all* x. To tell Maple that this is meant to be an identity valid for all x, we use the syntax `solve(identity(a*(x-1)^2+b*(x+1)^2=x,x),{a,b})`. Similarly, to find u such that $sin(t + u) = cos(t)$ for all t, enter `solve(identity(sin(t+u)=cos(t),t),{u})`.

8. Standard functions

(8.1) Most functions can be entered using their usual names, for example `sin(x)` or `ln(y)`.

(8.2) However, you must always use brackets: enter `tan(x)` instead of `tan x`, and `sin(2*x)` instead of `sin 2x`.

(8.3) The absolute value of x, traditionally written as $|x|$, must be entered as `abs(x)`.

(8.4) Both `ln(x)` and `log(x)` refer to the natural logarithm (to base e) of x.

(8.5) If you want to use the logarithm to base 10 then you should enter `log[10](x)`. In many cases you will actually need to enter `simplify(log[10](x))` to get a useful answer. Similarly, enter `log[2](x)` for $\log_2(x)$ (the logarithm to base 2) and so on.

(8.6) The function e^x should be entered as `exp(x)`.

(8.7) The hyperbolic function $\sinh(x) = (e^x - e^{-x})/2$ can be entered as `sinh(x)`, and similarly for the functions $\cosh(x) = (e^x + e^{-x})/2$ and $\tanh(x) = \sinh(x)/\cosh(x)$. Maple is not as good as it should be at dealing with these functions; it often works better to write expressions explicitly in terms of e^x instead.

(8.8) The traditional notation $\sin^2(x)$ refers to the square of $\sin(x)$, which could also be written $(\sin(x))^2$ or $\sin(x)^2$. In Maple we suggest that you enter this as `sin(x)^2`. It will not work to enter `sin^2(x)`. Similar comments apply to $\tan^2(x)$ and so on. Note that $\sin(x^2)$ means something different again: it is the sin of the square of x, not the square of the sin.

(8.9) In traditional notation $1/\sin(x)$ is sometimes written as $\csc(x)$ or $\mathrm{cosec}(x)$. In Maple `csc(x)` will work, but `cosec(x)` will not. Of course, you can also just enter `1/sin(x)`.

(8.10) In traditional notation $\tan^{-1}(x)$ or $\arctan(x)$ refers to the angle θ such that $\tan(\theta) = x$. This is completely different from the number $\tan(x)^{-1} = 1/\tan(x)$. In Maple you should enter `arctan(x)` (**not** `tan^(-1)(x)` or anything like that). Similar comments apply to $\arcsin(x)$, $\mathrm{arctanh}(x)$ and so on.

(8.11) When $x \geq 0$, the notation `sqrt(x)` means, by definition, the positive square root of x. (There is more to say if x is negative or complex, but we pass over that here.) Thus, for example, `sqrt(9)` is definitely 3 and not -3.

9. Defining your own functions

(9.1) If you want to define $f(x) = x^2$, enter `f := (x) -> x^2;` Note that the arrow symbol is typed as a dash (`-`) followed by a greater-than sign (`>`), and is not related to the arrow keys used to move your cursor.

(9.2) It **does not** work very well to enter `f(x):=x^2;` instead. If you do this, then Maple will convert $f(x)$ to x^2 (when the argument is just the letter x) but it will not convert $f(y)$ to y^2, or $f(3)$ to 9.

(9.3) Similarly, to define the sequence $a(n) = (-1)^n/n$, enter `a := (n) -> (-1)^n/n;`, not `a(n):=(-1)^n/n;`.

(9.4) It **does not** work to use notation like `f'(x)` for the derivative of $f(x)$; see Section 11 instead.

(9.5) You can define a function of several variables in a similar way. For example, to define $g(a, u, v) = u^a + v^a$, enter `g := (a,u,v)->u^a+v^a`.

(9.6) Suppose you define g as above, and later on you want a reminder of the definition. You should enter `print(g)`, and Maple will print out out $(a, u, v) \to u^a + v^a$. For reasons that we will not explore here, it does not work to enter `g` instead of `print(g)`.

10. Plotting

(10.1) To plot the graph $y = x^3 - x$ from $x = -2$ to $x = 2$ (for example), enter `plot(x^3-x,x=-2..2);`. Note that there is no "$y =$" in the plot command. Similarly, to plot $t/(e^t - 1)$ from $t = -1$ to $t = 3$, enter `plot(t/(exp(t)-1),t=-1..3);`.

(10.2) To make the vertical axis run from 0 to 10 (for example), enter `plot(x^3-x,x=-2..2,0..10);`. Note that you must enter `0..10`, **not** `y=0..10` here. The horizontal range (`x=-2..2`) includes the name of the variable, but the vertical range does not.

(10.3) To make the vertical scale the same as the horizontal scale, add the option `scaling=constrained` (eg `plot(x^3-x,x=-2..2,scaling=constrained);`).

(10.4) If the function to be plotted is complicated, it may be convenient to give the definition as a separate command, for example `y:=t^2+t^3+t^5+t^7+t^11+t^13;` and then `plot(y,t=0..1);` rather than just `plot(t^2+t^3+t^5+t^7+t^11+t^13,t=0..1)`.

(10.5) To plot the graph $y = x^3 + x$ in the same picture, enter `plot([x^3-x,x^3+x],x=-2..2);`. (Note that the outer pair of brackets are round, and the inner pair are square. Note also the different placing of brackets between here and **[10.8]** below.) Similarly, to plot $\cos(t)$ and $\sin(t)$ together from $t = -3\pi$ to $t = 3\pi$, enter `plot([sin(t),cos(t)],t=-3*Pi..3*Pi)`.

(10.6) To find and skip over discontinuities, add the option `discont=true`; for example:

```
plot(frac(x),x=-3..3,discont=true);
```

(10.7) To improve the accuracy of a graph, use the option `numpoints=1000`, for example `plot(x*sin(Pi/x),x=-1..1,numpoints=1000);`. (The number 1000 should always give a good picture, but may slow things down. Any number greater than 50 will improve the default picture.)

(10.8) To plot the curve given parametrically by $x = 1/(1 + t^2)$ and $y = \sin(t)$ from $t = -5$ to $t = 5$, enter

```
plot([1/(1+t^2),sin(t),t=-5..5]);
```

Similarly, to plot the curve $(x, y) = (\sin(t), \cos(2t))$ for $t = 0$ to 2π, enter `plot([sin(t),cos(2*t),t=0..2*Pi])`. The option `scaling=constrained` is often useful in this kind of plot.

(10.9) Note carefully the placing of the square brackets above. If you enter `plot([1/(1+t^2),sin(t)],t=-5..5);` instead, you get the graphs $y = 1/(1 + t^2)$ and $y = \sin(t)$ drawn together in the same picture, which is something rather different.

(10.10) In a parametric plot, if you want the x-axis to run from -3 to 3 and the y-axis from -2 to 2, you should enter `plot([1/(1+t^2),sin(t),t=-5..5],view=[-3..3,-2..2]);`.

(10.11) To plot a curve given implicitly, say by the equation $x + y + x^2y^2 = 1$, enter

```
plots[implicitplot](x+y+x^2*y^2=1,x=-3..3,y=-3..3);
```

Similarly, you can plot the circle $x^2 + y^2 = 4$ like this:

```
plots[implicitplot](x^2+y^2=4,x=-2..2,y=-2..2);
```

The option `scaling=constrained` is often useful in this kind of plot.

(10.12) Implicit plots will rarely give a good picture unless you tell Maple to start with a finer grid, which you can do like this:

```
plots[implicitplot](x+y+x^2*y^2=1,x=-3..3,y=-3..3,grid=[200,200]);
```

The number 200 generally seems to work well, but you could try a larger or smaller one.

(10.13) To generate and save a plot for future use, enter something like this: `MyPicture:=plot(x^2+x,x=2..4):`. Note that this ends with a colon, to prevent Maple from printing out vast lists of coordinates. You can then display the picture by entering `MyPicture;` (with a semicolon).

(10.14) You can display two pictures together by entering `plots[display](MyPicture,MyOtherPicture);` (for example). This can also be done without saving the pictures separately. For example, you can plot $y = x^3$ together with $x^2 + y^2 = 1$ like this: `plots[display](plot(x^3,x=-1..1),plots[implicitplot](x^2+y^2=1,x=-1..1,y=-1..1));`

(10.15) You can use `plots[listplot](...)` to plot a list of points. For example `plots[listplot]([8,7,5,1])` plots four points with y coordinates 8, 7, 5 and 1, and x coordinates 1, 2, 3 and 4. Similarly, `plots[listplot]([9,9,9,9,9,9,9])` plots seven points, all at height 9, at x-coordinates $1, \ldots, 7$.

(10.16) By default, `listplot` draws a line joining the specified points. If you just want the points themselves, use the option `style=POINT`, for example `plots[listplot]([8,7,5,1],style=POINT)`

(10.17) If you want, you can specify the x-coordinates as well, rather than just taking them to be $1, 2, 3, \ldots$. For example,
`plots[listplot]([[1,1],[1,-1],[-1,1],[-1,-1]])` plots the four points with coordinates $(1, 1)$, $(1, -1)$, $(-1, 1)$ and $(-1, -1)$, which are the four corners of a square.

(10.18) If you enter `with(plots):`, then Maple will let you enter `implicitplot(...)` instead of `plots[implicitplot](...)`, `listplot(...)` instead of `plots[listplot](...)` and `display(...)` instead of `plots[display](...)`, and so on. The same applies to various other commands that we have not yet discussed.

(10.19) If you restart Maple for any reason, you will have to enter `with(plots):` again afterwards.

11. Differentiation

(11.1) To differentiate $x + \sin(x)$ with respect to x (for example), enter `diff(x+sin(x),x)`. This is equivalent to $\frac{d}{dx}(x + \sin(x))$ in traditional notation. Similarly, to differentiate $1/(1 - t^2)$ with respect to t we enter `diff(1/(1-t^2),t)`.

(11.2) Note that it is necessary to specify the variable with respect to which you want to differentiate. If you enter `diff(x^2,x)` you get $2x$, but if you just enter `diff(x^2)`, you get an error message.

(11.3) To differentiate y three times with respect to x (for example), enter `diff(y,x,x,x)` or `diff(y,x$3)`. This is equivalent to d^3y/dx^3 in traditional notation. Similarly, to find the second derivative of $\sin(\theta)$ with respect to θ we enter `diff(sin(theta),theta$2)`.

(11.4) In a different kind of traditional notation, you could define $f(t) = t^3 + 4$ (for example), and then $f'(t)$ would mean the derivative $\frac{d}{dt}f(t) = 3t^2$. The Maple equivalent would be to enter `f:=(t)->t^3+4;` for the definition, and `D(f)(t)` for the derivative. Note that `D(f)(t)` means exactly the same as `diff(f(t),t)`, but it is sometimes more convenient.

(11.5) If x and y are related implicitly by an equation (say $x + y + x^2y^2 = 1$) then you can find dy/dx in terms of x and y by implicit differentiation, like this:

```
implicitdiff(x+y+x^2*y^2=1,y,x)
```

(11.6) Sometimes you just want to mention the derivative without actually working it out. You can do this by writing `Diff` instead of `diff`. For example, if you enter `Diff(x^3+4,x)` then Maple will print out $\frac{d}{dx}(x^3 + 4)$; if you then enter `value(%)`, Maple will convert this to $3x^2$. Similarly, `Diff(t^10,t,t)` prints as $\frac{d^2}{dt^2}t^{10}$, whereas `value(Diff(t^10,t,t))` or `diff(t^10,t,t)` gives $90t^8$.

(11.7) To find the Taylor series of a function $f(x)$ near a point $x = a$ to order n, enter `series(f(x),x=a,n)`. Note that "order n" means that the highest term allowed is a multiple of x^{n-1}, and all terms x^n and higher are discarded. For example, `series(1/x,x=2,7)` gives the Taylor series for $1/x$ near $x = 2$ including terms up to $(x - 2)^6$.

(11.8) The `series` command gives an answer including a term like $O(x^7)$, to indicate that there are infinitely many more terms, starting with a multiple of x^{12}. Maple will not let us do very much with this answer until we have converted it to an ordinary polynomial (without the $O(x^7)$ term on the end). To do this, use the command `convert(...,polynom)`.

12. Integration

(12.1) To calculate an indefinite integral like $\int \tan(x)\,dx$, enter `int(tan(x),x)`.

(12.2) Maple's answer will never include an arbitrary constant '$+c$'. In some contexts (particularly, solution of differential equations) the constant makes a difference and must be included. In other contexts it is irrelevant; Maple cannot help you to decide about this.

(12.3) To calculate a definite integral like $\int_1^5 t^3\,dt$, enter `int(t^3,t=1..5)`. Similarly, to calculate $\int_0^2 \sqrt{1+x}\,dx$, enter `int(sqrt(1+x),x=0..2)`.

(12.4) It is also possible to have infinite limits. For example, $\int_{-\infty}^{\infty}(1 + x^4)^{-1}dx$ can be entered as

```
int((1+x^4)^(-1),x=-infinity..infinity)
```

(12.5) As with differentiation, it is sometimes useful to mention an integral without evaluating it. You can do this by writing `Int` instead of `int`. For example, if you enter `Int(x^2,x)` then Maple will print it as $\int x^2\,dx$; if you then enter `value(%)`, then Maple will convert it to $x^3/3$.

13. Sequences and sums

(13.1) To generate a sequence of terms, use the `seq()` command. For example, the sequence $1/2, 1/4, 1/6, 1/8, 1/10$ is the sequence of terms $1/(2k)$ for $k = 1, 2, 3, 4, 5$; it can be generated by the command `seq(1/(2*k),k=1..5)`.

(13.2) To find the sum of a sequence of terms, use the `sum()` command. For example, $\frac{1}{2}+\frac{1}{4}+\frac{1}{6}+\frac{1}{8}+\frac{1}{10}$ is the sum of the terms $1/(2k)$ for k from 1 to 5, or in other words $\sum_{k=1}^{5} 1/(2k)$. This can be entered as `sum(1/(2*k),k=1..5)`.

(13.3) Occasionally it works better to use the `add()` command instead of `sum()`. We will not explore the difference here.

14. Some common problems

(14.1) Remember `*`'s for multiplication.
- Remember especially that Maple converts `2(x+y)` to 2: you probably meant $2 * (x + y)$ instead.
- Enter xy as `x*y`; Maple will print this as $x\ y$, with a little space between the letters. If you see xy with no space, that means that Maple is referring to a single variable with the two-letter name "xy". You probably typed `xy` instead of `x*y` by mistake.
- For the product of b and $x + y$, enter `b*(x+y)`; Maple will print this as $b\ (x + y)$, with the b in italics and followed by a little space. If you see $\mathrm{b}(x + y)$ (with an upright b and no space) then Maple is referring to the value of a function called b at $x + y$ (just as $\mathrm{f}(x + y)$ is the value of the function f at $x + y$). You probably typed `b(x+y)` instead of `b*(x+y)`, by mistake.

(14.2) Remember not to use `*`'s for application of functions. Things like $\sin x$ in traditional notation must be entered as `sin(x)`, not `sin*x` or `sin x`.

(14.3) $\frac{a+b}{c+d}$ must be entered as `(a+b)/(c+d)` (not `a+b/c+d`), and $\frac{a}{(u+v)(x+y)}$ must be entered as `a/((u+v)*(x+y))` (not `a/(u+v)*(x+y)`).

(14.4) $a^{1/4}$ must be entered as `a^(1/4)`, not `a^1/4`.

(14.5) Always use round brackets (not square or curly ones) for algebraic grouping. Square brackets are used for lists or for subscripts (e.g. x_5 is `x[5]`). Curly brackets are used for sets.

(14.6) Remember to remove definitions when you have finished with them, by entering the `restart;` command. If you enter an expression like $(x+y)^5$ and Maple converts it to something completely unrelated like $(\sin(\theta)+e^{-u})^5$, the most likely reason is that you previously defined `x:=sin(theta);` and `y:=exp(-u);`, and you forgot to remove these definitions.

(14.7) If you give x a value and then try to use it as a plotting variable (e.g. by entering `plot(x^2,x=-1..1);`) you will get the cryptic message "Error, (in plot) invalid arguments". You should restart (or just enter `unassign('x');`) and try again.

(14.8) Remember the constant $e = 2.71828\ldots$ must be entered as `exp(1)`, and e^x must be entered as `exp(x)`. Maple will print this as $\mathbf{e}^x$, with a bold e. If Maple is doing funny things with exponentials, look at the e's in your expression. If they are not bold, then you must have entered e^x as `e^x` instead of `exp(x)`, by mistake. You can cure this by entering `e:=exp(1);`.

(14.9) Remember that the constant $\pi = 3.14159\ldots$ must be entered as `Pi`, not `pi`. If Maple is doing funny things like refusing to simplify $\sin(\pi)$ to 0, it may be that you entered `pi` instead of `Pi` by mistake. You can cure this by entering `pi:=Pi;`.

15. Why the Classic Worksheet?

There are two different user interfaces to Maple: the "Classic Worksheet" interface (with the orange icon) and the "Standard" interface (with the red icon). We will use the Classic interface, for several reasons:

- The syntax used by the Classic interface is similar to that used in other contexts: programming languages such as Java or PHP, spreadsheet systems such as Excel, mathematical software such as Matlab, and so on. It is therefore a good general skill to be familiar with this syntax. By contrast, the skills that you need for the Standard interface are of little use for other systems.
- With the Classic interface you type a sequence of characters and they appear on the screen as you typed them. This means that it is easy for the lecturer to explain what you need to type, and how to fix the most obvious problems. The Standard interface involves much more clicking with the mouse and typing cursor and tab keys that are not directly visible on the screen. This makes it much harder to provide comprehensible written instructions.
- The online tests also use the same syntax as the Classic interface (and it is not possible to change that).
- The Classic interface is much faster on less powerful computers.

You are free to experiment with the Standard interface if you wish to, but you should remember that some points in the exam will rely on knowledge of the Classic interface.